



# Excel 2016 Math with Dates and Times





# Microsoft Excel 2016: Math with Dates and Times

1.5 hour

In this advanced math workshop, we will unlock the secrets of date/time math in Excel; learn lots of shortcuts and custom formatting options; explore many of the built in date/time functions; build our own date/time equations. By the end of this workshop you will be able to create a simple timesheet, and calculate how many days between two dates, excluding holidays and weekends. This advanced workshop assumes prior experience with Microsoft Excel; experience with building equations in Excel required.

| Dates in Excel  | .1  |
|---|-----|
| Times in Excel  | . 1 |
| Useful Date/Time Shortcuts  | . 1 |
| Custom Formatting   | . 2 |
| Days, Months, and Years   | . 2 |
| Hours, minutes, and seconds   | . 2 |
| Totaling Time   | . 3 |
| Simple Date/Time Math   | .3  |
| Date Worksheet Functions  | . 3 |
| DATE - Returns the sequential serial number that represents a particular date         | . 3 |
| DATEVAULE - Returns the serial number of a date                                       | . 4 |
| DAY - Returns the numeric value of the day in a valid date                            | . 4 |
| DAYS360 - returns the number of days between two dates based on a 360-day year        | .4  |
| EDATE - Returns serial number of the date that is a number of months away from a date | . 5 |
| EOMONTH - Returns serial number for the last day of the month                         | . 5 |
| MONTH- Returns the numeric value of the month in a valid date                         | . 5 |
| TODAY - the serial number of the current date   | . 5 |
| NETWORKDAYS - returns the number of whole working days between two dates              | .6  |
| WEEKDAY - Returns the day of the week corresponding to a date                         | . 6 |
| WEEKNUM - Returns the day of the week corresponding to a date                         | . 6 |
| WORKDAY - Returns a date that is a number of working days before or after a date      | .7  |
| YEAR- Returns the numeric value of the year in a valid date                           | . 7 |
| YEARFRAC - Returns fraction of the year of the number of whole days between two dates | .7  |
| Time Functions  | .8  |
| HOUR - Returns the hour of a time value   | . 8 |
| MINUIE- Returns the hour of a time value  | . 8 |
| NOW - the serial number of the current date and time                                  | . 8 |
| SECOND - Returns the Seconds of a time value  | . 8 |
| TIME - Returns the sequential serial number that represents a particular time         | .9  |
| TIMEVAULE - RETURNS THE SERIAL NUMBER OF A TIME                                       | .9  |
| Datedit Function  | .9  |
| Class Exercise  | 10  |



Pandora Rose Cowart Education/Training Specialist UF Health IT Training

C3-013 Communicore PO Box 100152 Gainesville, FL 32610-0152 (352) 273-5051 prcowart@ufl.edu http://training.health.ufl.edu

# Dates in Excel

If you've ever lost the date formatting on a cell, you have seen it turn into a strange number. For example, according to Excel these cells are the same number:



The serial number 41, 564 tells us how many days it has been since January 1st, 1900\*. The date serial numbers are sequential, one day at a time:

|   | А          | В     |
|---|------------|-------|
| 1 | 10/17/2013 | 41564 |
| 2 | 10/18/2013 | 41565 |
| 3 | 10/19/2013 | 41566 |

\*Note: The default date system for the Macintosh begins with January 1, 1904. This may cause some confusion if you try to use the same file in both a Mac and a PC. The setting can be changed in the Excel Options on the File menu.

#### <u>Times in Excel</u>

One day has 24 hours, so in Microsoft Excel, 1 is equivalent to 24 hours, 0.5 is equivalent to 12 hours. If we take our October 17th date, and add in a time of 12:00pm, it translates into 41564.5, the 0.5 representing the half-way point of the day.

1=24 hours, 0.5=12 hours, 0.25=6 hours...



If you leave the date off a time, Excel will default to 1/0/1900 as the 'understood' date. You can ignore it, but realize that is what happens if you change a time format into a date/time format. All three of these cells contain 12:00 PM, they are just displayed with different formats:

|   | А        | В        | С               |
|---|----------|----------|-----------------|
| 1 | 12:00 PM | 1/0/1900 | 1/0/00 12:00 PM |

#### Useful Date/Time Shortcuts

| Shortcut     | Result                               | Note                       |
|--------------|--------------------------------------|----------------------------|
| Ctrl-;       | Current Date                         | Control Semicolon          |
| Ctrl-:       | Current Time                         | Control Colon              |
| Shift-Ctrl-3 | Formats the cell to show DD-MMM-YYYY | 3/# from the full keyboard |
| Shift-Ctrl-2 | Formats the cell to show h:mm AM/PM  | 2/@ from the full keyboard |

# Custom Formatting

#### Days, Months, and Years

You can format a cell with a preset list of options in the Format Cells Window.

- Click the More button in the Number group
- Right-click on a cell and choose Format cells
- Select the cell and press Ctrl-1 to open this window

Excel has a pretty extensive list of date and time formats, but it is possible to custom build a date format, using simple abbreviations.

| Number  | Alignment       | Font   | Border                                      | Fill       | Protection                   |            |
|---|-----------------|--|---|------------|------------------------------|------------|
| Category:   |                 |  |   |            |                              |            |
| General A<br>Number<br>Currency                                 |                 | ≜ Sam  | ple   |            |                              |            |
| Date  | 2               | Type:  | Province -                                  |            |                              | 205        |
| Time<br>Percentage<br>Fraction<br>Scientific<br>Text<br>Special |                 | *Wed<br>3/14<br>3/14/<br>03/14<br>14-M<br>14-M | dnesday, Marc<br>/01<br>#/01<br>ar<br>ar-01 | h 14, 2001 | -                            | -          |
| Costoni   |                 | Locale   | (location):                                 |            |                              |            |
|   |                 | Englis   | sh (U.S.)                                   |            |                              |            |
| Date form   | ats display dat | * te and time s                                | serial numbers                              | as date v  | alues. Date formats that beg | in with an |

| Dates for Tuesday, February 3, 2004 |         |       |          |      |      |  |
|-------------------------------------|---------|-------|----------|------|------|--|
| Day                                 |         | Month |          | Year |      |  |
| D                                   | 3       | m     | 2        | 10/  | 04   |  |
| dd                                  | 03      | mm    | 02       | уу   | 04   |  |
| ddd                                 | Tue     | mmm   | Feb      | 1000 | 2004 |  |
| dddd                                | Tuesday | mmmm  | February | уууу | 2004 |  |

#### Hours, minutes, and seconds

If you use "m" immediately after the "h" or "hh" code or immediately before the "ss" code, Excel displays minutes instead of the month.

| Times for 1:02:05 |    |         |    |         |    |
|-------------------|----|---------|----|---------|----|
| Hours             |    | Minutes |    | Seconds |    |
| Н                 | 1  | m       | 2  | S       | 5  |
| hh                | 01 | mm      | 02 | SS      | 05 |

If you would like to use the 12 hour clock you need to add the appropriate designator at the end. Access will accept any of the following: AM/PM; am/pm; A/P; a/p; AMPM.

| Times for 17:02:05 |      |               |       |               |          |  |
|--------------------|------|---------------|-------|---------------|----------|--|
| Hours              |      | 24 hour clock |       | 12 hour clock |          |  |
| Н                  | 17   | h:m           | 17:2  | h:m am/pm     | 5:2 PM   |  |
| hh                 | 17   | h:mm          | 17:02 | h:mm am/pm    | 5:02 PM  |  |
| h AM/PM            | 5 PM | hh:mm         | 17:02 | hh:mm am/pm   | 05:02 PM |  |

If you need to go smaller, adding .00 after the second format (ss.00) will give the fraction of a second.

### Totaling Time

The time formats in Excel stay within the defaults of 24 hours, 60 minutes, 60 seconds. If we do math we may want it to display the times beyond these boundaries. For this, we use the brackets [] around the abbreviation.

| Beyond the Boundaries |    |            |                 |            |    |  |
|-----------------------|----|------------|-----------------|------------|----|--|
| 36 Hours              |    | 75 minutes |                 | 75 seconds |    |  |
| h                     | 12 | m          | (assumes month) | S          | 15 |  |
| hh                    | 12 | mm         | (assumes month) | SS         | 15 |  |
| [h]                   | 36 | [m]        | 75              | [s]        | 75 |  |

### Simple Date/Time Math

Because dates and times are stored as numbers you can do simple math with them.

| Begin      | End        | Formula                 | Result                                      |
|------------|------------|-------------------------|---|
| 01/02/2013 | 05/15/2013 | =Last Date – First Date | Days Elapsed = 133                          |
| 8:15 AM    | 5:00 PM    | =Last Time – First Time | Time Elapsed = 8:45<br>(8 hours 45 minutes) |

Times are a fraction of a day. As mentioned earlier, 0.5=12 hours. If we multiply a time by 24 we should get the numeric value.

| Time    | Time * 24 |
|---------|-----------|
| 8:45 AM | 6:00 PM   |

Why doesn't it look right? You have to format the result as a true number field.

| Time    | Time * 24 (formatted as a number) |  |
|---------|-----------------------------------|--|
| 8:45 AM | 8.75                              |  |

Remember if you are going beyond the 24 hour clock, you will need to set up a custom format.

#### **Date Worksheet Functions**

Adapted from Excel Help

#### DATE - Returns the sequential serial number that represents a particular date

Syntax: DATE(year, month, day)

*Year* The value of the year argument can include one to four digits.

Month A positive or negative integer representing the month of the year from 1 to 12. If month is greater than 12, month adds that number of months to the first month in the year specifiedDay A positive or negative integer representing the day of the month from 1 to 31.

| Equation               | Result     | Notes  |
|------------------------|------------|--|
| =DATE(2009, 3, 15)     | 03/15/2009 |  |
| =DATE(5, 10, 15)       | 10/15/1905 | Notice this counts year from 1900, thus will not assume "2005" |
| =DATE(2010, 15,<br>20) | 03/20/2011 | Notice this is going into the following year because 15 months |
| =DATE(1976, 11,<br>35) | 120/5/1976 | Notice this is going into the following month because 35 days  |
| =DATE(2007, -1, 5)     | 11/05/2006 | Negative values will go backwards in time, minus day/year      |

# DATEVAULE - Returns the serial number of a date

Converts a date that is stored as text to a serial number that Excel recognizes as a date. To view a date serial number as a date, you must apply a date format to the cell.

# Syntax: DATEVALUE(date\_text)

date\_text represents a date between 01/01/1900 and 12/31/9999 saved as text

| Equation                 | Result  | Notes                   |
|--------------------------|---------|-------------------------|
| =DATEVALUE(3/15/2009)    | #VALUE! | Date not in Text format |
| =DATEVALUE("10/15/1905") | 2115    |                         |
| =DATEVALUE("3/20/2011")  | 40622   |                         |
| =DATEVALUE(A1)           | #VALUE! | A1 = 12/5/1976          |
| =DATEVALUE(A1)           | 28099   | A1 = "12/5/1976"        |

# DAY - Returns the numeric value of the day in a valid date

### Syntax: =DAY(serial number)

Serial Number is the date of the day you are trying to find.

| Equation           | Result | Notes   |
|--------------------|--------|---|
| =DAY(3/15/2009)    | 0      | Serial Number not a Date (Excel sees 3÷15÷2009) |
| =DAY("10/15/1905") | 15     |   |
| =DAY(40622)        | 20     | Equivalent to 3/20/2011                         |
| =DAY(A1)           | 5      | A1 = 12/5/1976                                  |

# DAYS360 - returns the number of days between two dates based on a 360-day year

360 day year assumes twelve 30-day months, this is often used with accounting calculations. If *start\_date* occurs after *end\_date*, the DAYS360 function returns a negative number.

# Syntax: DAYS360(start\_date, end\_date, [method])

Start Date and End Date are valid dates which represent the starting and ending dates. Method Optional logical value that specifies whether to use the U.S. (NASD) or European method in the calculation. False or omitted will give us the U.S. method, True will use the European method.

Remarks: If *start\_date* is the last day of the month, both methods set the start\_date to the 30th of the month, but if the end\_date is the last day of the month the US method will change the date to the 1st of the next month, the European will change the date to the 30th.

| Equation         | Result | Notes                            |
|------------------|--------|----------------------------------|
| =DAYS360(A1, A2) | 8      | A1 = 02/27/2009, A2 = 03/05/2009 |
| =DAYS360(B1, B2) | 1      | B1 = 01/30/2009, B2 = 02/01/2009 |
| =DAYS360(C1, C2) | 1      | C1 = 01/31/2009, C2 = 02/01/2009 |
| =DAYS360(D1, D2) | 360    | D1 = 11/15/2009, D2 = 11/15/2010 |

# EDATE - Returns serial number of the date that is a number of months away from a date

Syntax: EDATE(start\_date, months)

Start Date is a valid date that represents the starting date

*Months* is the number of months before or after start\_date. A positive value for months yields a future date; a negative value yields a past date

| Equation        | Result |            | Notes           |
|-----------------|--------|------------|-----------------|
| =EDATE(A1, 1)   | 39899  | 03/27/2009 | A1 = 02/27/2009 |
| =EDATE(A1, 10)  | 40174  | 12/27/2009 | A1 = 02/27/2009 |
| =EDATE(A1, 100) | 42913  | 06/27/2017 | A1 = 02/27/2009 |
| =EDATE(A1, -1)  | 39840  | 01/27/2009 | A1 = 02/27/2009 |

### EOMONTH - Returns serial number for the last day of the month

EOMonth => End of month

### Syntax: EOMONTH(start\_date, months)

Start Date is a valid date that represents the starting date

*Months* number of months before or after *start\_date*. A positive value for months yields a future date; a negative value yields a past date.

| Equation          | Result |            | Notes           |
|-------------------|--------|------------|-----------------|
| =EOMONTH(A1, 1)   | 39903  | 03/31/2009 | A1 = 02/27/2009 |
| =EOMONTH(A1, 10)  | 40178  | 12/31/2009 | A1 = 02/27/2009 |
| =EOMONTH(A1, 100) | 42916  | 06/30/2017 | A1 = 02/27/2009 |
| =EOMONTH(A1, -1)  | 39844  | 01/31/2009 | A1 = 02/27/2009 |

# MONTH- Returns the numeric value of the month in a valid date

Syntax: =MONTH(serial number)

Serial Number is the date of the month you are trying to find.

| Equation             | Result | Notes                   |
|----------------------|--------|-------------------------|
| =MONTH("10/15/1905") | 10     |                         |
| =MONTH(40622)        | 3      | Equivalent to 3/20/2011 |
| =MONTH(A1)           | 12     | A1 = 12/5/1976          |

# TODAY - the serial number of the current date

Syntax: TODAY()

| Equation | Result    | Notes                                 |  |
|----------|-----------|---------------------------------------|--|
| =TODAY() | 12/5/2016 | This will always be the current date. |  |

# NETWORKDAYS - returns the number of whole working days between two dates

Syntax: NETWORKDAYS(start\_date, end\_date, holidays)

*Start Date* and *End Date* are valid dates.

Holidays is an optional range of one or more dates to exclude from the working calendar. The list can be either a range of cells that contains the dates or an array constant of the serial numbers that represent the dates

| Equation                    | Result | Notes   |
|-----------------------------|--------|---|
| =NETWORKDAYS(A1, A2)        | 5      | A1 = 02/27/2009, A2 = 03/05/2009                                      |
| =NETWORKDAYS(B1, B2)        | 1      | B1 = 01/30/2009, B2 = 02/01/2009                                      |
| =NETWORKDAYS(C1, C2)        | 12     | C1 = 07/01/2009, C2 = 07/15/2009                                      |
| =NETWORKDAYS(C1, C2, C3)    | 11     | C1 = 07/01/2009, C2 = 07/15/2009,<br>C3 = 07/04/1999                  |
| =NETWORKDAYS(D1, D2, D3:D4) | 259    | D1 = 11/15/2009, D2 = 11/15/2010, D3 =<br>01/01/2010, D4 = 07/04/2010 |

### WEEKDAY - Returns the day of the week corresponding to a date

Syntax: =WEEKDAY(serial number, return\_type)

Serial Number is the date of the day you are trying to find.

Return Type is a number that determines the type of return value. 1 or omitted sees the week as 1-Sunday, 7-Saturday. 2 sees the week as 1-Monday, 7-Sunday. 3 sees the week as 0-Monday, 6-Sunday.

| Equation               | Result | Notes                           |
|------------------------|--------|---------------------------------|
| =WEEKDAY("12/15/1976") | 4      | Wednesday, December 15, 1976    |
| =WEEKDAY(40622)        | 1      | Sunday, March 20, 2011          |
| =WEEKDAY(A1)           | 7      | A1 = Saturday, October 28, 1905 |

#### WEEKNUM - Returns the day of the week corresponding to a date

Syntax: =WEEKNUM(serial number, return\_type)

Serial Number is the date of the day you are trying to find.

Return Type is a number that determines the type of return value. 1 - Week begins on Sunday.
Weekdays are numbered 1 through 7; 2 - Week begins on Monday. Weekdays are numbered 1 through 7.

Remarks: The WEEKNUM function considers the week containing January 1 to be the first week of the year.

| Equation                 | Result | Notes                           |
|--------------------------|--------|---------------------------------|
| =WEEKNUM("12/15/1976") 5 |        | Wednesday, December 15, 1976    |
| =WEEKNUM(40622)          | 13     | Sunday, March 20, 2011          |
| =WEEKNUM(A1)             | 43     | A1 = Saturday, October 28, 1905 |

# WORKDAY - Returns a date that is a number of working days before or after a date

Syntax: WORKDAY(start\_date, days, holidays)

*Start Date* is a valid date that represents the starting date.

*Days* is the number of non-weekend and non-holiday days before or after start\_date. A positive value yields a future date; a negative value, a past date.

Holidays is an optional range of one or more dates to exclude from the working calendar. The list can be either a range of cells that contains the dates or an array constant of the serial numbers that represent the dates

| Equation                | Result |              | Notes   |
|-------------------------|--------|--------------|---|
| =WORKDAY(A1, 1)         | 39874  | Mon 3/2/09   | A1 = 02/27/2009                                       |
| =WORKDAY(A1, 10)        | 39885  | Fri 3/13/09  | A1 = 02/27/2009                                       |
| =WORKDAY(A1, 100)       | 40011  | Fri 7/17/09  | A1 = 02/27/2009                                       |
| =WORKDAY(C1, 100, C3)   | 40135  | Wed 11/18/09 | C1 = 07/01/2009,<br>C2 = 07/04/2009                   |
| =WORKDAY(D1, D2, D3:D4) | 40274  | Tue 4/6/10   | D1 = 11/15/09, D2 = 100,<br>D3 = 1/1/10, D4 = 1/18/10 |

# YEAR- Returns the numeric value of the year in a valid date

Syntax: =YEAR(serial number)

Serial Number is the date of the year you are trying to find.

| Equation            | Result | Notes                   |
|---------------------|--------|-------------------------|
| =YEAR("10/15/1905") | 5      |                         |
| =YEAR(40622)        | 2011   | Equivalent to 3/20/2011 |
| =YEAR(A1)           | 1976   | A1 = 12/5/1976          |

### **YEARFRAC - Returns fraction of the year of the number of whole days between two dates** Syntax: YEARFRAC(start\_date, end\_date, basis)

Start Date and End Date are valid dates which represent the starting and ending dates Basis is the type of day count basis to use

0-US/NASD 30/360; 1-Actual/Actual; 2-Actual/360; 3-Actual/365; 4-European 30/360

| Equation          | Result      | Notes                         |
|-------------------|-------------|-------------------------------|
| =YEARFRAC(A1, A2) | 1           | A1 = 1/1/2009, A2 = 1/1/2010  |
| =YEARFRAC(A1, A3) | 1.013888889 | A1 = 1/1/2009, A3 = 1/1/2010  |
| =YEARFRAC(A1, A4) | 0.455555556 | A1 = 1/1/2009, A4 = 6/15/2009 |
| =YEARFRAC(A1, A5) | 0.538888889 | A1 = 1/1/2009, A5 = 7/15/2009 |
| =YEARFRAC(A1, A6) | 13.53888889 | A1 = 1/1/2009, A6 = 7/15/2022 |

# **Time Functions**

#### HOUR - Returns the hour of a time value

Syntax: HOUR(serial\_number)

*Serial\_Number* the time that contains the hour you want to find.

Remarks: Times may be entered as text strings within quotation marks (for example, "6:45 PM"), as decimal numbers (for example, 0.78125, which represents 6:45 PM), or as results of other formulas or functions (for example, TIMEVALUE("6:45 PM")).

| Equation  | Result | Notes               |
|-----------|--------|---------------------|
| =HOUR(A1) | 20     | A1 = 8:28 PM        |
| =HOUR(B1) | 8      | B1 = 8:28 AM        |
| =HOUR(C1) | 15     | C1 = 15:43:12       |
| =HOUR(D1) | 17     | D1 = 1/2/2003 17:52 |

#### MINUTE- Returns the hour of a time value

#### Syntax: MINUTE (serial\_number)

*Serial\_Number* the time that contains the hour you want to find.

Remarks: Times may be entered as text strings within quotation marks (for example, "6:45 PM"), as decimal numbers (for example, 0.78125, which represents 6:45 PM), or as results of other formulas or functions (for example, TIMEVALUE("6:45 PM")).

| Equation    | Result | Notes               |
|-------------|--------|---------------------|
| =MINUTE(A1) | 28     | A1 = 8:28 PM        |
| =MINUTE(C1) | 43     | C1 = 15:43:12       |
| =MINUTE(D1) | 52     | D1 = 1/2/2003 17:52 |

**NOW** - the serial number of the current date and time Syntax: NOW( )

| Equation | Result         | Notes                                      |  |  |  |  |  |
|----------|----------------|--|--|--|--|--|--|
| =NOW()   | 3/7/2009 11:02 | This will always be the current date/time. |  |  |  |  |  |

#### SECOND - Returns the Seconds of a time value

Syntax: SECOND(serial\_number)

*Serial\_Number* the time that contains the hour you want to find.

Remarks: Times may be entered as text strings within quotation marks (for example, "6:45 PM"), as decimal numbers (for example, 0.78125, which represents 6:45 PM), or as results of other formulas or functions (for example, TIMEVALUE("6:45 PM")).

| Equation    | Result | Notes                  |  |  |
|-------------|--------|------------------------|--|--|
| =SECOND(A1) | 0      | A1 = 8:28 PM           |  |  |
| =SECOND(C1) | 12     | C1 = 15:43:12          |  |  |
| =SECOND(D1) | 15     | D1 = 1/2/2003 17:52:15 |  |  |

# TIME - Returns the sequential serial number that represents a particular time

Syntax: TIME(hour, minute, second)

- *Hour* is a number from 0 (zero) to 32767 representing the hour. Any value greater than 23 will be divided by 24 and the remainder will be treated as the hour value
- *Minute* is a number from 0 to 32767 representing the minute. Any value greater than 59 will be converted to hours and minutes.
- Second is a number from 0 to 32767 representing the second. Any value greater than 59 will be converted to hours, minutes, and seconds.

| Equation          | Result      | Notes                         |
|-------------------|-------------|-------------------------------|
| =TIME(15, 3, 15)  | 3:03:20 PM  |                               |
| =TIME(0, 0, 2000) | 12:33:20 AM | 2000 seconds = 33 min, 20 sec |
| =TIME(C1, C2, C3) | 6:12:09 AM  |                               |

# TIMEVAULE - Returns the serial number of a time

Converts a time that is stored as text to a serial number that Excel recognizes as a time. To view a time serial number as a time, you must apply a time format to the cell.

### Syntax: TIMEVALUE(time\_text)

time\_text represents a time between 01/01/1900 and 12/31/9999 saved as text

| Equation                   | Result      | Notes                      |
|----------------------------|-------------|----------------------------|
| =TIMEVALUE("3:03:20 PM ")  | 0.627314815 |                            |
| =TIMEVALUE("12:33:20 AM ") | 0.023148148 |                            |
| =TIMEVALUE(A1)             | #VALUE!     | A1 = 6:12:09 AM (not text) |
| =TIMEVALUE(A1)             | 0.2584375   | A1 = "6:12:09 AM "         |

# **Datedif Function**

There is a "hidden" function that can find the difference between two dates and return different increments. When you subtract two dates you get the number of days between them. The Datedif function is not in the help files or list of formats, but is probably one of the most powerful date related worksheet functions.

Syntax: DATEDIF(Begin Date, End Date, "Interval")

|   | Δ          | R          | C                   | Interval | Description                                   |
|---|------------|------------|---------------------|----------|---|
| 1 | Begin Date | End Date   | C                   | D        | Number of Days                                |
| 2 | 1/2/2013   | 5/15/2015  |                     | М        | Number of Months                              |
| 3 |            |            |                     | Y        | Number of Years                               |
| 4 | Result     | Difference | Formula             | YM       | Number of months, not counting years          |
| 5 | 863        | Days       | =DATEDIF(A2,B2,"d") | YD       | Number of days not counting years             |
| 6 | 28         | Months     | =DATEDIF(A2,B2,"m") |          | Number of days, not counting years            |
| 7 | 2          | Years      | =DATEDIF(A2,B2,"y") | MD       | Number of days, not counting years and months |

#### **Class Exercise**

#### **Open "Dates-1-FormatsFills"**

- 1) Sheet "Shortcuts"
  - B2: Enter a date month and day only (7/4)
  - B3: Enter a date with the year (7/4/2016)
  - B2: Enter a date with the year (7/4/2016)
    - Format doesn't change
  - Select B2:B3
    - From the Number Format drop down, choose Short Date
    - From the Number Format drop down, choose Long Date
    - Press Shift-Ctrl-3 to Quick-format
    - Format Cells (More Numbers) view date formats
  - B6: Press Ctrl-Semicolon (Ctrl-;)
  - B7: Press Ctrl-Colon (Ctrl-:)
  - B8: Press Ctrl-Semicolon, press a <u>space</u>, Press Ctrl-Colon
  - D6: =B6
  - D7: =B7
  - D8: =B8
  - Select D6:D8
  - Format number to GENERAL
  - Dates How many days since 1/1/1900
  - Times fraction of a 24 hour clock
  - B11: =Today()
  - B12: =Now()
    - Reformat to only show time

- 2) Sheet "Date Formats"
  - Select A3:F3
  - Press Ctrl-; and then Press Ctrl-Enter
    - Today's date should appear in all the cells
  - Select each cell and set a custom format
    - B3: DDD
    - C3: DDDD
    - D3: MMM
    - D4: MMMM
    - D4: DDDD, MMMM D, YYYY
- 3) Sheet "V-Date Pattern"
  - Select A3:C3, Drag fill handle to row 10
  - Select D3:E3, Drag fill handle to row 10
  - Select F3:G3, Drag fill handle to row 10
- 4) Sheet "H-Date Pattern"
  - Select B1:B7
  - Drag Fill handle to Column G
- 5) Sheet "AutoFill Days"
  - Select A2:D2, Enter today's date
    - Ctrl-; then Ctrl-Enter
  - Use Fill handle on each column one at a time
    - Select A2, Fill to Row 10
    - Select B2, Fill to Row 10, change fill options to By Weekday
    - Select C2, Fill to Row 10, change fill options to By Month
    - Select D2, Fill to Row 10, change fill options to By Year
  - In F2 and F3 put the next two pay days
    - Select <u>both</u> dates and fill to row 10

#### **Open "Dates-2-ApptDates"**

- 1) Sheet "Past Appts"
  - Simple Math, Reformat to General as needed
    - D3: =today()-A3
    - E3: =( today()-A3 )/30
    - F3: =( today()-A3 )/365
  - DateDif function
    - See Page 9 for details
    - G3: =datedif(A3, today(), "D")
    - H3: =datedif(A3, today(), "M")
    - I3: =datedif(A3, today(), "Y")
- 2) Sheet "Future Appts"
  - D2: =A2+5
  - E2: =Workday(A2, 5)
    - See Page 6 for details
- 3) Sheet "End and Begin Month"
  - EOMonth function
    - See Page 5 for details
    - B2: =EOMonth(A2, 1)
    - C2: = EOMonth(A2, 2)
    - D2: =EOMonth(A2, 1) + 1

#### Open "Dates-3-Holidays.xlsx"

- 1) Sheet "Holidays"
  - This list was created by the instructor of this workshop. We are going to use the whole column A in our equations. If you create your own list, please keep in mind, the list can ONLY include dates, NO TITLE.

- 2) Sheet "Predict End Date", (reformat results as dates, if needed)
  - C2: =A2 + B2
  - D2: =Workday(A2, B2)
  - E2: = Workday (A2, B2, Holidays!A:A)
  - AutoFill values for all dates
- 3) Sheet "# of Days"
  - C2: =B2-A2
  - D2: =NetworkDays(A2, B2)
  - E2: =NetworkDays(A2, B2, Holidays!A:A)
  - AutoFill values for all dates
- 4) See Page 6 for details on WORKDAY and NETWORKDAY

#### **Open** " Times-1-FormatsFills.xlsx"

- 1) Sheet "Shortcuts"
  - B2: Enter a number and a colon
  - B3: Enter a number between 1 and 12, a space, then an A or P
  - Select B2:B3
    - From the Number Format drop down, choose Time
    - Press Shift-Ctrl-2 to Quick-format
    - Format Cells (More Numbers) view Time formats
  - B6: Press Ctrl-Semicolon (Ctrl-;)
  - B7: Press Ctrl-Colon (Ctrl-:)
  - B8: Press Ctrl-Semicolon, press a <u>space</u>, Press Ctrl-Colon
  - B11: =Today()
  - B12: =Now()

- 2) Sheet "Time Formats"
  - Select B3:E3
    - B3:E3 Type "1:23 a" and press Ctrl-Enter
    - B4:E4 Type "1:23 p" and press Ctrl-Enter
    - B5:E5 Type "36:00" and press Ctrl-Enter
  - Custom Format
    - B3:B5 h:mm am/pm
    - C3:C5 hh:mm
    - D3:D5 hh:mm:ss am/pm
    - E3:E5 [hh]:mm
      - 1. Brackets to BREAK IT out of the 24-hour clock
- 3) Sheet "AutoFill Times"
  - A2: 9 a AutoFill to Row 10
  - B2: 9 p AutoFill to Row 10
  - C2: 6:00
  - C3: 12:00
  - Select C2:C3
  - AutoFill to row 10
    - Reformat to go beyond the 24 hour clock [hh]:mm
  - D2: 0:15
  - D3: 0:30
  - Select D2:D3
  - AutoFill to row 10
  - E2: =D2\*24
    - Multiply by 24 to mathematically move it out of a time
    - Reformat to a comma format, AutoFill to row 10

#### Open "Times-2-TimeSheet.xlsx"

- 1) Reported time is Time Out minus Time In
  - a. F2: =(C2-B2) + (E2-D2)
  - b. AutoFill down to Row 6
- 2) Total time is the sum of all the reported times
  - a. F7: AutoSum
  - b. Reformat for 24-hour clock (brackets)
- 3) Hours Worked should be linked to Total
  - a. C10: =F7
  - b. C12: =C10\*F7
    - i. Answers wrong! Remember 40:00 is not 40 hours
  - c. C10: =F7\*24
  - d. Reformat C10 to a comma style
- 4) Exit without saving, and then try to do it again.

|    | А            | В           |     | С      |       | D         | E     |       | F        |
|----|--------------|-------------|-----|--------|-------|-----------|-------|-------|----------|
| 1  | Date         | In          | Out |        |       | In        |       | Out   | Reported |
| 2  | Mon          | 8:00        | 1   | 12:00  |       | 12:30     |       | 16:45 | 8:15     |
| 3  | Tue          | 8:15        | 1   | 11:15  | 12:30 |           | 17:30 |       | 8:00     |
| 4  | Wed          | 8:30        | 1   | 12:45  |       | 13:30 17: |       | 17:15 | 8:00     |
| 5  | Thu          | hu 7:00     |     | 12:00  | 12:30 |           | 17:00 |       | 9:30     |
| 6  | Fri          | Fri 7:45    |     | 14:00  |       |           |       |       | 6:15     |
| 7  |              |             |     |        |       |           |       | Total | 40:00    |
| 8  |              |             |     |        |       |           |       |       |          |
| 9  |              |             |     |        |       |           |       |       |          |
| 10 | Hours Worked |             |     | 40.00  |       |           |       |       |          |
| 11 |              | Hourly Rate | \$  | 15.00  |       |           |       |       |          |
| 12 | 2 Total Due  |             | \$  | 600.00 |       |           |       |       |          |